A SIMULATION OF A TRANSIENT ALTERNATING RENEWAL PROCESS

By

George Marcellious Scherrer

United States Naval Postgraduate School

THESIS

A SIMULATION OF A TRANSIENT
ALTERNATING RENEWAL PROCESS

by

George Marcellious Scherrer, Jr.

Thesis Advisor: R. W. Butterworth

March 1971

A Simulation of a Transient Alternating Renewal Process

by

George Marcellious Scherrer, Jr.
Captain, United States Army
B.S., United States Military Academy, 1963

Submitted in partial fulfillment of the
requirements for the degree of

MASTER OF SCIENCE IN OPERATIONS RESEARCH

from the

NAVAL POSTGRADUATE SCHOOL
March 1971

ABSTRACT

This thesis describes an alternating renewal process and a computer program that simulates a transient alternating renewal process. The purpose of the simulation is to test analytical approximations for the efficiency of a system described by such a process.

TABLE OF CONTENTS

I. DESCRIPTION OF AN ALTERNATING RENEWAL PROCESS

This thesis describes an alternating renewal process and gives a computer program that simulates a transient alternating renewal process. Consider a system that can be in one of two states, up or down. The system starts up and alternates between up and down, remaining in each state a random time. The successive times in each state are jointly independent. The sequence of up times are identically distributed with distribution F_u and the sequence of down times are identically distributed with distribution F_d. X_i and Y_i are the corresponding i^{th} duration of time the system is in the up state and down state. The means of the two distributions are assumed to be finite and are denoted by μ_u and μ_d respectively. The up rate and down rate are denoted by λ_u and λ_d, and are related to the means as

$$\begin{aligned} \mu_u &= \frac{1}{\lambda_u} = \text{mean time up} \\ \mu_d &= \frac{1}{\lambda_d} = \text{mean time down} \end{aligned} \qquad (1)$$



Figure 1. An Alternating Renewal Process.

Two quantities of interest concerning an alternating renewal process are the availability and the efficiency. The availability, A(t), is the probability that the system will be up at time t. The efficiency, E(t), is the ratio of the expected amount of up time during [0,t] to t. The relationship between E(t) and A(t), which will be derived later, is

$$E(t) = \frac{\int_0^t A(x)\,dx}{t} \tag{2}$$

An alternating renewal process has two imbedded renewal processes. Reference 1 gives a detailed description of a renewal process. One process, call it a type 1 process, has as its renewals the events when the system goes from the up to the down state. This is a general renewal process with the first interevent distribution $F_1 = F_u$ and the remaining interevent distributions all being $F = F_u \circ F_d$, where 'o' denotes convultion. The duration of the i^{th} interevent time is $Y_{i-1} + X_i$ for $i \geq 2$. A second imbedded renewal process has as its renewals the events when the system goes up. The common interevent distribution is $F = F_u \circ F_d$. This will be denoted as a type 2 process.

The expected number of renewals during [0,t] is denoted by M(t) and is called the renewal function. If F is a function of bounded variation, that is, the difference of two monotone functions, and vanishes to the left of zero, then



Figure 2. Type 1 and Type 2 Renewals and the Interevent Times.

$$F^*(s) = \int_{0-}^{\infty} \exp(-sx)\,dF(x)$$

is defined as the Laplace-Stieljes Transform of F. If the times between renewals are distributed as $F_1, F, \ldots,$ then

$$M^*(s) = \frac{F_1^*(s)}{1 - F^*(s)} . \tag{3}$$

For a derivation of this expression see Ref. 1.

For the type 1 imbedded renewal process with renewal function $M_1(t)$,

$$M_1^*(s) = \frac{F_u^*(s)}{1 - F_u^*(s)\, F_d^*(s)} .$$

For the type 2 imbedded renewal process with renewal function $M_2(t)$,

$$M_2^*(s) = \frac{F_u^*(s)\, F_d^*(s)}{1 - F_u^*(s)\, F_d^*(s)} .$$

Define the random variable S(t) such that

$$S(t) = \begin{cases} 1 & \text{if the system is up at } t \\ 2 & \text{if the system is down at } t. \end{cases}$$



Figure 3. Alternating Renewal Process with Two Imbedded Renewal Processes.

Let N_t^i = number of renewals of type i, i = 1,2, that occur in [0,t]. Then

$$S(t) = N_t^2 - N_t^1 + 1.$$

If the expectation is taken, then

$$\begin{aligned} E[S(t)] &= E[N_t^2] - E[N_t^1] + 1 \\ &= M_2(t) - M_1(t) + 1 = A(t) \end{aligned} \tag{4}$$

The Laplace-Stieljes Transform of (4) gives

$$A^*(s) = M_2^*(s) - M_1^*(s) + 1$$

$$= \frac{F_u^*(s)\ F_d^*(s)}{1 - F_u^*(s)\ F_d^*(s)} - \frac{F_u^*(s)}{1 - F_u^*(s)\ F_d^*(s)} + 1$$

$$A^*(s) = \frac{1 - F_u^*(s)}{1 - F_u^*(s)\ F_d^*(s)} \quad . \tag{5}$$

Let the random variable Q(t) be the amount of up time during [0,t] and R(t) be the expected amount of up time over the interval [0,t]. Then

$$Q(t) = \int_0^t S(x)\, dx \, .$$

If the expectation is taken, then

$$R(t) = E[Q(t)] = E[\int_0^t S(x)dx]$$

$$= \int_0^t A(x)\, dx$$

from which (2) immediately follows

$$E(t) = \frac{\int_0^t A(x)\, dx}{t} = R(t)/t \, .$$

If (5) can be inverted, then availability and, hence efficiency, can be obtained. For example, if the distributions of the up and down times are exponential with corresponding rates λ_u and λ_d, then

$$\left.\begin{aligned} F_u(t) &= 1 - \exp(-\lambda_u t) \\ F_d(t) &= 1 - \exp(-\lambda_d t) \\ F_i^*(s) &= \frac{\lambda_i}{\lambda_i + s} \qquad i = u,d \end{aligned}\right\} \qquad (6)$$

gives

$$A^*(s) = \frac{1 - \left(\frac{\lambda_u}{\lambda_u+s}\right)}{1 - \left(\frac{\lambda_u}{\lambda_u+s}\right)\left(\frac{\lambda_d}{\lambda_d+s}\right)}$$

$$= \frac{\lambda_d + s}{\lambda_u + \lambda_d + s}$$

Inverting yields

$$A(t) = \frac{\lambda_d}{\lambda_d+\lambda_u} + \left(\frac{\lambda_u}{\lambda_u+\lambda_d}\right)\left(\exp[-(\lambda_u+\lambda_d)t]\right)$$

Use of (2) gives

$$E(t) = \frac{\int_0^t \left\{ \frac{\lambda_d}{\lambda_d+\lambda_u} + \left(\frac{\lambda_u}{\lambda_u+\lambda_d}\right)\left(\exp[-(\lambda_u+\lambda_d)x]\right)\right\} dx}{t}$$

$$= \frac{\lambda_d}{\lambda_u+\lambda_d} + \frac{\lambda_u}{t(\lambda_u+\lambda_d)^2}\left(1 - \exp[-(\lambda_u+\lambda_d)t]\right)$$

$A^*(s)$ is not always easy to invert. For instance, if the down distribution is k-erlang, then

$$F_d^*(s) = \left(\frac{\lambda_d}{\lambda_d+s}\right)^k$$

$$A^*(s) = \frac{1 - \frac{\lambda_u}{\lambda_u+s}}{1 - \left(\frac{\lambda_u}{\lambda_u+s}\right)\left(\frac{\lambda_d}{\lambda_d+s}\right)^k}$$

$$= \frac{\lambda_u(\lambda_d+s)}{(\lambda_u+s)(\lambda_d+s)^k - \lambda_u\lambda_d^k} \qquad (7)$$

The inversion of (7) can be obtained in principle but it is not easily expressed in closed form. If the down distribution is log normal, a reasonable assumption for many systems, then A(t) cannot be obtained in closed form.

If the system has been operating for a long period of time it will settle down into a steady state. For large t an asymptotic approximation for E(t) can be found using

$$\lim_{t \to \infty} \frac{\int_0^t A(x)\, dx}{t} = \frac{1}{n!} \lim_{s \to 0+} s^{n-1} A^*(s)$$

and

$$\left.\frac{d\, F^*(s)}{d\, s}\right|_{s=0} = -\mu = \left. F^{*\prime}(s)\right|_{s=0}, \quad F \text{ is any distribution.}$$

Letting n = 1, then

$$\lim_{t \to \infty} \frac{\int_0^t A(x)\, dx}{t} = \frac{1}{1!} \lim_{s \to 0+} s^0 \left\{ \frac{1 - F_u^*(s)}{1 - F_u^*(s)\, F_d^*(s)} \right\} .$$

Use of L'Hôspital's rule for limits in the indeterminate form gives

$$\lim_{t \to \infty} \frac{\int_0^t A(x)\, dx}{t} = \lim_{s \to 0+} \frac{-F_u^{*\prime}(s)}{-[F_u^{*\prime}(s) F_d^*(s) + F_u^*(s) F_d^{*\prime}(s)]}$$

$$= \frac{\mu_u}{\mu_u + \mu_d} \tag{8}$$

The derivation of (8) was independent of the type of distributions. If the system is in a transient state, and equilibrium has not been reached, then the asymptotic approximation is not valid and other approximations must be made. Hence, a need exists for testing the approximations. The program described in the next section simulates an alternating renewal process for this purpose.

II. A SIMULATION OF AN ALTERNATING RENEWAL PROCESS

This section describes a simulation of an alternating renewal process. Three distributions are used.

Exponential

$$F(t) = 1 - \exp(-\lambda t) \qquad t \geq 0$$
$$= 0 \qquad \text{otherwise.}$$

k-erlang

$$F(t) = 1 - \sum_{i=0}^{k-1} \frac{(\beta t)^i \exp(-\beta t)}{i!} \qquad t \geq 0$$
$$= 0 \qquad \text{otherwise}$$

Log normal

$$F(t) = \int_0^t \frac{1}{\sigma x\sqrt{2\pi}} \exp\left[-\frac{1}{2\sigma^2} (\ell n\ x-a)^2\right] dx \qquad t \geq 0$$
$$= 0 \qquad \text{otherwise.}$$

Two combinations of the distributions are used. In one combination the up times are distributed exponentially and the down times are distributed k-erlang. In the other situation, the up distribution is the same but the down distribution is log normal.

R(t), the expected amount of up time, and E(t), the efficiency of the system over the interval [0,t], are the output parameters and are provided for ten time intervals. Analytical approximations for E(t) are given for comparison.

The input parameters vary with the choice of down distribution. There are some inputs common to both situations: Δt the unit time interval, λ_u the up time rate, λ_d the down time rate, the number of simulations and runs desired. If the k-erlang distribution is selected the parameter k must be provided. If the log normal distribution is used the variance must be given. The computer program gives more information on the inputs.

The program functions generally as outlined in the flowchart. Each simulation has these basic steps. First the up cycle is entered. While in the up cycle and upon exiting, checks are made to determine if a time period has been spanned. If not, the next step is the down cycle. While in the down cycle and upon exiting, tests are made again to determine whether or not the end of a time period has been reached. If not, the program returns to the up cycle. If at any check the time interval over which the simulation is to occur has been spanned, and, if the desired number of simulations has not been completed, then another simulation begins. After the required number of simulations have been completed, the output parameters are computed and printed.

The program was written in FORTRAN IV and used on an IBM 360/67 computer.


START
READ INPUT
INITIALIZE ARRAYS
1
UP CYCLE
IS T SPANNED
Yes
ALL SIMULATIONS COMP.
Yes
A
No
No
1
DOWN CYCLE
IS T SPANNED
Yes
ALL SIMULATIONS COMP.
Yes
A
No
No
1
1


A
COMPUTE
R(t) AND
E(t)
ANALYTICAL
APPROXIMATION
OF E(t)
PRINT
OUTPUT
ANY
MORE
RUNS
Yes
START
No
STOP

CASE	I			II			III		
[0,t]	R(t)	E(t)	E#(t)	R(t)	E(t)	E#(t)	R(t)	E(t)	E#(t)
0.4	0.39421	0.98552	0.98241	0.39239	0.98098	0.98241	0.39488	0.98720	0.98241
0.8	0.77379	0.96724	0.96883	0.77170	0.96262	0.96883	0.78158	0.97697	0.96883
1.2	1.13795	0.94829	0.95823	1.14016	0.95013	0.95823	1.15991	0.96659	0.95823
1.6	1.49920	0.93700	0.94988	1.50856	0.94285	0.94988	1.52824	0.95515	0.94988
2.0	1.85931	0.92965	0.94323	1.87633	0.93816	0.94323	1.89176	0.94588	0.94323
2.4	2.21729	0.92387	0.93788	2.23982	0.93326	0.93788	2.25400	0.93916	0.93788
2.8	2.57228	0.91867	0.93354	2.60057	0.92877	0.93554	2.61522	0.93400	0.93554
3.2	2.93088	0.91590	0.92997	2.96064	0.92520	0.92997	2.97634	0.93010	0.92997
3.6	3.29383	0.91495	0.92701	3.32148	0.92263	0.92701	3.33812	0.92725	0.92701
4.0	3.65569	0.91392	0.92454	3.68389	0.92097	0.92454	3.70015	0.92503	0.92454

Case I - Log normal down distribution, variance = 0.3066

II - Four -erlang down distribution

III - Exponential down distribution

Down Rate = 0.9, Up rate = 0.1

Asymptotic E(t) = 0.9

E#(t) = the first analytical approximation of E(t) and is the exact data if both distributions are exponential.

```
C     THIS PROGRAM SIMULATES A TRANSIENT ALTERNATING RENEWAL PROCESS. THE
C SYSTEM STARTS UP. THE UP DISTRIBUTION IS EXPONENTIAL WITH PARAMETER RLAMB.
C THE DCWN DISTRIBUTION IS EITHER LOG NORMAL WITH PARAMETERS DOWNRT AND DOWNVR,
C OR , K-ERLANG WITH PARAMETERS DOWNRT AND KERLG.
C
C INPUTS:
C          TDELT-   THE INCREMENTAL TIME JUMP. TEN JUMPS ARE MADE STARTING AT 0
C          RLAMB-   UP TIME RATE
C          LDIST-   PARAMETER USED TO INDICATE TYPE OF DOWN DISTRIBUTION
C                   0 = LOG NORMAL
C                   1 = K-ERLANG
C          KERLG-   EXTERNAL PARAMETER THAT GIVES A VALUE FOR K IN THE K-ERLANG,
C                   INTERNALLY SET = 0 FOR LOG NORMAL
C          DOWNRT-  COWN TIME RATE
C          DOWNVR-  VARIANCE OF DOWN DISTRIBUTION COMPUTED INTERNALLY FOR K-ERLANG
C          NSIM-    THE NUMBER OF SIMULATIONS
C          NRUN-    THE NUMBER OF RUNS
C          IREAD-   A PARAMETER INDICATING THE NUMBER OF DATA SETS
C OUTPUTS:
C          EXPT-    THE AVERAGE UP TIME PER INTERVAL STARTING FROM 0
C          EFCY-    AVERAGE EFFICIENCY PER INTERVAL STARTING FROM 0
C          APROX-   ANALYTICAL APPROXIMATION OF EFFICIENCY
C          DOWNRT-  COWN RATE
C          DOWNVR-  DOWN VARIANCE
C
      DIMENSION UPT(10),SUMUPT(10),DNT(10),TPER(10),EXPT(10),EFCY(10),
     1APROX(4,10)
      NAMELIST/CNSTS/TDELT,RLAMB,LDIST,KERLG,DOWNRT,DOWNVR,NSIM,NRUN,
     1IREAD
      IDUMMY=0
      DO 1 LEVEL=1,1000
    1 X=RN(0)
    5 READ(5,CNSTS)
C
C INITIALIZE TIME VECTOR AND ARRAY USED TO STORE UP TIMES
C
      TPER(1)=TDELT
      DO 10 J=2,10
      TPER(J)=TPER(J-1)+TDELT
   10 CONTINUE
C
C CHECK TO SEE TYPE OF DOWN DIST. 0- LOG NORMAL . 1- K- ERLANG.
C
      IF(LDIST.GE.1) GO TO 70
C
C LCG NORMAL  COMPUTE MEAN AND VARIANCE OF NORMAL DIST.
C
      KERLG=0
```

```
      DWNMEN=1.0/DOWNRT
      UVAR=ALOG((DOWNVR/(DWNMEN**2))+1.0)
      UMEAN=(ALOG(DWNMEN))-(UVAR/2.0)
      UDEV=SQRT(UVAR)
      GO TO 72
C
C K- ERLANG DIST. COMPUTE VARIANCE
C
   70 BETA=DOWNRT*KERLG
      DCWNVR=1.0/(KERLG*(DOWNRT**2))
C
C START NRUN RUNS
C
   72 DO 2000 MANY=1,NRUN
      DO 75 NIL=1,10
      SUMUPT(NIL)=0.0
   75 CONTINUE
C
C START NSIM SIMULATIONS
C
  100 DO 1100 LOOP=1,NSIM
      TTIM1=0.0
      TTIM2=0.0
      ACTU=0.0
      ACTD=0.0
      TIMU=0.0
      TIMD=0.0
      DO 110 IN=1,10
      UPT(IN)=0.0
      DNT(IN)=0.0
  110 CONTINUE
      I=1
C
C THIS IS THE UP CYCLE
C
  200 X=RN(0)
      TIMU=(-ALOG(X)/RLAMB)
      TTIM2=TIMU+TTIM1
      IF(TPER(I)-TTIM2)400,500,600
C
C TIME PERIOD EXCEEDED
C
  400 ACTU=TPER(I)-TTIM1+ACTU
  410 UPT(I)=ACTU
      DNT(I)=ACTD
      IF(I.GE.10) GO TO 1000
      I=I+1
      IF(TPER(I)-TTIM2)420,430,440
```

```
C
C TIME PERIOD EXCEEDED AGAIN
C
  420 ACTU=ACTU+TDELT
      GO TO 410
C
C TIME PERIOD EXCEEDED BEFORE EQUALLED NOW
C
  430 ACTU=ACTU+TDELT
      UPT(I)=ACTU
      DNT(I)=ACTD
      IF(I.GE.10) GO TO 1000
      I=I+1
      TTIM1=TTIM2
      GO TO 300
C
C TIME PERIOD EXCEEDED BEFORE,NOW UP TIME LESS
C
  440 ACTU=ACTU+TTIM2-TPER(I-1)
      UPT(I)=ACTU
      TTIM1=TTIM2
      GO TO 300
C
C TIME PERIOD EQUALLED
C
  500 ACTU=ACTU+TIMU
      UPT(I)=ACTU
      DNT( I)=ACTD
      TTIM1=TTIM2
      IF(I.GE.10) GO TO 1000
      I=I+1
      GO TO 300
C
C TIME PERIOD NOT EXCEEDED OR EQUALLED
C
  600 ACTU=ACTU+TIMU
      TTIM1=TTIM2
      UPT(I)=ACTU
      GO TO 300
C
C START DOWN CYCLE
C
C CHECK TO SEE TYPE OF DOWN DISTRIBUTION
C
  300 IF(LDIST.GE.1) GO TO 320
C
C LCG NORMAL DOWN DIST.
C
```

```
      X=(GRN(IDUMMY)*UDEV)+UMEAN
      TIMD=EXP(X)
      GO TO 350
C
C K-ERLANG DIST.
C
  320 TIMD=0.0
      DO 330 MN=1,KERLG
      Y=RN(0)
      TIMD=(-ALOG(Y)/BETA)+TIMD
  330 CONTINUE
  350 TTIM2=TIMD+TTIM1
      IF(TPER(I)-TTIM2)700,800,900
C
C TIME PERIOD EXCEEDED
C
  700 ACTD=TPER(I)-TTIM1+ACTD
  710 UPT(I)=ACTU
      DNT(I)=ACTD
      IF(I.GE.10) GO TO 1000
      I=I+1
      IF(TPER(I)-TTIM2) 720,730,740
C
C TIME PERIOD EXCEEDED AGAIN
C
  720 ACTD=ACTD+TDELT
      GO TO 710
C
C TIME PERIOD EXCEEDED BEFORE NOW EQUALLED
C
  730 ACTD=ACTD+TDELT
      DNT(I)=ACTD
      UPT(I)=ACTU
      IF(I.GE.10) GO TO 1000
      I=I+1
      TTIM1=TTIM2
      GO TO 200
C
C TIME PERIOD EXCEEDED BEFORE NOW DOWN TIME LESS
C
  740 ACTD=ACTD+TTIM2-TPER(I-1)
      DNT(I)=ACTD
      TTIM1=TTIM2
      GO TO 200
C
C TIME PERIOD EQUALLED
C
  800 ACTD=ACTD+TIMD
```

```
      DNT(I)=ACTD
      UPT(I)=ACTU
      TTIM1=TTIM2
      IF(I.GE.10) GO TO 1000
      I=I+1
      GO TO 200
C
C TIME PERIOD NOT EXCEEDED OR EQUALLED
C
  900 ACTD=ACTD+TIMD
      TTIM1=TTIM2
      DNT(I)=ACTD
      GO TO 200
 1000 DO 1010 ITOT=1,10
      SUMUPT(ITOT)=UPT(ITOT)+SUMUPT(ITOT)
 1010 CONTINUE
 1100 CONTINUE
C
C CCMPUTE AVERAGE UP TIME AND EFFICIENCY FOR EACH  TIME INTERVAL
C
      DO 1600 I=1,10
      EXPT(I)=0.0
      EFCY(I)=0.0
      EXPT(I)=SUMUPT(I)/NSIM
      EFCY(I)=EXPT(I)/TPER(I)
 1600 CONTINUE
C
C CCMPUTE ANALYTICAL APPROXIMATIONS
C
      SUMRTE=RLAMB+DOWNRT
      COEF1=DOWNRT/SUMRTE
      COEF2=RLAMB/(SUMRTE**2)
      DO 1700 I=1,10
      APROX(1,I)=COEF1+(COEF2*((1-EXP(-SUMRTE*TPER(I)))/TPER(I)))
 1700 CONTINUE
C
C PRINT OUTPUTS
C
      WRITE(6,6002)
 6002 FORMAT('1',43X,'THIS IS AVERAGE UP TIME IN INTERVAL')
      WRITE(6,6003) (EXPT(I),I=1,10)
 6003 FORMAT('0',4X,10F12.7)
      WRITE(6,6004)
 6004 FORMAT('0',41X,'THIS IS AVERAGE EFFICIENCY PER INTERVAL')
      WRITE(6,6005)(EFCY(I),I=1,10)
 6005 FORMAT('0',4X,10F12.7)
      WRITE(6,6006)
 6006 FORMAT('0',39X,'THIS IS THE FIRST ANALYTICAL APPROXIMATION')
```

```
      WRITE(6,6007) (APROX(1,I),I=1,10)
 6007 FORMAT('0',4X,10F12.7)
      WRITE(6,6008) DOWNRT,DOWNVR
 6008 FORMAT('0',6X,'DOWN RATE=',F12.7,3X,'VARIANCE=',F12.7///)
      WRITE(6,CNSTS)
 2000 CONTINUE
      IREAD=IREAD-1
      IF(IREAD.GT.0) GO TO 5
      STOP
      END
```

BIBLIOGRAPHY

1. Cox, D. R., Renewal Theory, Methuen and Co., Ltd., 1962.

2. Shoonan, Martain L., Probabilistic Reliability: An Engineering Approach, McGraw-Hill, 1968.

3. Butterworth, R. W., Lecture Notes for Stochastic Models, Course presented at Naval Postgraduate School, Monterey, California, July-September, 1970.

INITIAL DISTRIBUTION LIST

		No. Copies
1.	Defense Documentation Center Cameron Station Alexandria, Virginia 22314	2
2.	Library, Code 0212 Naval Postgraduate School Monterey, California 93940	2
3.	Department of Operations Analysis, Code 55 Naval Postgraduate School Monterey, California 93940	1
4.	Assistant Professor R. W. Butterworth, Code 55Bd Department of Operations Analysis Naval Postgraduate School Monterey, California 93940	1
5.	Captain George M. Scherrer, Jr. Box 266 Shawneetown, Illinois 62984	1

UNCLASSIFIED

Security Classification

DOCUMENT CONTROL DATA - R & D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall report is classified)

1. ORIGINATING ACTIVITY *(Corporate author)* Naval Postgraduate School Monterey, California 93940	2a. REPORT SECURITY CLASSIFICATION Unclassified 2b. GROUP

3. REPORT TITLE

A SIMULATION OF A TRANSIENT ALTERNATING RENEWAL PROCESS

4. DESCRIPTIVE NOTES *(Type of report and, inclusive dates)*

Master's Thesis; March 1971

5. AUTHOR(S) *(First name, middle initial, last name)*

George Marcellious Scherrer, Jr., Captain, United States Army

6. REPORT DATE	7a. TOTAL NO. OF PAGES	7b. NO. OF REFS
March 1971	25	3

8a. CONTRACT OR GRANT NO. b. PROJECT NO. c. d.	9a. ORIGINATOR'S REPORT NUMBER(S) 9b. OTHER REPORT NO(S) *(Any other numbers that may be assigned this report)*

10. DISTRIBUTION STATEMENT

Approved for public release; distribution unlimited.

11. SUPPLEMENTARY NOTES	12. SPONSORING MILITARY ACTIVITY
	Naval Postgraduate School Monterey, California 93940

13. ABSTRACT

This thesis describes an alternating renewal process and a computer program that simulates a transient alternating renewal process. The purpose of the simulation is to test analytical approximations for the efficiency of a system described by such a process.

DD FORM 1 NOV 65 1473 (PAGE 1)

S/N 0101-807-6811

UNCLASSIFIED

Security Classification

A-31408

14. KEY WORDS	LINK A		LINK B		LINK C	
	ROLE	WT	ROLE	WT	ROLE	WT
ALTERNATING RENEWAL PROCESS						
RELIABILITY						
AVAILABILITY						
EFFICIENCY						

Thesis 125911
S333 Scherrer
c.1 A simulation of a
transient alternating
renewal process.